Physique Numérique (SMP6)

"Calcul formel" Maple

TP3 : Résolution d'équations, continuité, dérivation, intégration

Rappel et compléments :

1) pour la définition d'une **fonction** sous Maple, la syntaxe est la suivante :

```
> f:=x->x^2+1;  # on a définit une fonction qui, à un réel x quelconque associe f(x)
```

$$f := x \to x^2 + 1 \quad (1)$$

En particulier, on peut calculer f(2), f(π), f(a) etc.

```
> f(2),f(Pi),f(a);
```

$$5, \pi^2 + 1, a^2 + 1 \quad (2)$$

CE QU'IL NE FAUT PAS FAIRE pour définir une fonction :

```
> restart;f:=x^2+1;  # on a définit un nombre réel (qui, certes, dépend de x)
```

$$f := x^2 + 1 \quad (3)$$

On vient ici de définir une *expression en x*, c'est-à-dire un *nombre réel, un résultat*, PAS UNE FONCTION. Une bonne manière de comprendre la distinction est la suivante : dans ce dernier exemple, pour Maple, x est figé, il ne varie pas. En particulier si on veut calculer f(2), on n'obtient pas de résultat satisfaisant :

```
> f(2);
```

$$x(2)^2 + 1 \quad (4)$$

Pour obtenir $2^2 + 1$, il faut définir "manuellement" x comme étant 2. Si on veut $5^2 + 1$, il faut "manuellement" définir x comme 5.

```
> x:=2; f;
```

$$x := 2$$

$$5 \quad (5)$$

```
> x:=5;f;
```

$$x := 5$$

$$26 \quad (6)$$

2) Autre avertissement : si je veux définir non plus une seule fonction f mais une *suite de fonctions*

f_n : la première idée est de définir ce qui suit

```
> restart; fn:=x->x^n+1;f2(x);
```

$$fn := x \rightarrow x^n + 1$$

$$f2(x) \tag{7}$$

Problème : Maple n'a pas compris la dépendance en n de "fn" : Pour Maple, le symbole "fn" à gauche de la définition n'est simplement que la juxtaposition de la lettre "f" et de la lettre "n" ; remplacer la chaine de caractères "fn" par toute autre chaine de caractère serait revenu au même !

Une bonne façon de faire est d'expliciter la dépendance en n de la suite fn, en définissant f comme dépendant de DEUX variables, x et n :

```
> f:=(x,n)->x^n+1; f(x,2);
```

$$f := (x, n) \rightarrow x^n + 1$$

$$x^2 + 1 \tag{8}$$

3) Une fonction utile lorsqu'on souhaite définir des fonctions par morceaux ; il s'agit de la fonction *piecewise* :

```
> ?piecewise
```

```
> f:=x->piecewise(x>=1,x+1,x<-1,-x, Pi);plot(f(x),x=-5..5,y=0..5,
  discont=true);
```

$$f := x \rightarrow piecewise(1 \leq x, x + 1, x < -1, -x, \pi)$$



NB : on notera l'utilisation de l'option discont=true dans plot pour masquer les traits parasites dans le graphe au niveau des discontinuités.

1. Résoudre des équations

1.1 Résoudre une équation à une inconnue

Pour résoudre une équation à une inconnue, on utilise la commande *solve* dont la syntaxe est *solve (equation, variable).*
L'équation est définie par une égalité =, à bien différencier des affectations, qui se font par : =.

Commençons par les **équations polynomiales**. Maple résout complètement sur **C** les équations polynomiales de degré <=3 et quelques équations de degré supérieur.

```
> restart;
  solve(x=1+1/x,x);
```

$$\frac{1}{2}+\frac{\sqrt{5}}{2}, \frac{1}{2}-\frac{\sqrt{5}}{2} \qquad \textbf{(9)}$$

Ici, Maple donne les solutions exactes et explicites de l'équation. Le type de résultat renvoyé par Maple est une *séquence*. On peut demander les valeurs approchées des solutions par :

```
> evalf(%);
```

$$1.618033988, -0.6180339880 \qquad \textbf{(10)}$$

Malheureusement, en degré supérieur ou égal à 4, cela ne se passe pas toujours aussi bien.
Essayons de résoudre l'équation : x^7-x^6+x^2-1=0.

```
> S:=solve(x^7-x^6+x^2-1=0,x);
```

$$S := 1, \mathit{RootOf}\left(\_Z^6+\_Z+1, \mathit{index}=1\right), \mathit{RootOf}\left(\_Z^6+\_Z+1, \mathit{index}=2\right), \mathit{RootOf}\left(\_Z^6+\_Z+1, \mathit{index}=3\right), \mathit{RootOf}\left(\_Z^6+\_Z+1, \mathit{index}=4\right), \mathit{RootOf}\left(\_Z^6+\_Z+1, \mathit{index}=5\right), \mathit{RootOf}\left(\_Z^6+\_Z+1, \mathit{index}=6\right) \qquad \textbf{(11)}$$

Parfois, lorsque Maple ne peut pas trouver de solution explicite, ou lorsque celle-ci est trop compliquée pour être utilisable, les solutions sont exprimées à l'aide de la fonction *RootOf* ("racine de"), qui "représente" toutes les racines à la fois et aucune en particulier. Factorisons le polynôme de départ.

```
> factor(x^7-x^6+x^2-1);
```

$$(x-1)\left(x^6+x+1\right) \qquad \textbf{(12)}$$

Une racine évidente est 1, les autres racines sont celles de x^6+x+1 (au nombre de 6), que Maple refuse d'expliciter. Cependant, ce n'est pas complètement un échec car :
* certaines fonctions de Maple sont capables de travailler avec *RootOf,* et on peut continuer alors à "faire des calculs" :

```
> alias(alpha=RootOf(x^6+x+1));
```

$$\alpha \qquad \textbf{(13)}$$

```
> alpha^8;
  simplify(alpha^8);
```

$$\alpha^8 \qquad \textbf{(14)}$$

$$-\alpha^2(1+\alpha)$$

* si l'équation ne comporte aucun paramètre, on peut obtenir des valeurs numériques approchées des racines :

```
> evalf(S);
```

$$1., 0.9454023333+0.6118366938\,I, -0.1547351445+1.038380754\,I, -0.7906671888+0.3005069203\,I, -0.7906671888-0.3005069203\,I, -0.1547351445-1.038380754\,I, \qquad \textbf{(15)}$$

$$0.9454023333 - 0.6118366938\,\mathrm{I}$$

La fonction *allvalues* permet parfois de "déterminer" toutes les racines désignées par le *RootOf :* quand c'est possible, Maple donne des expressions exactes.

```
> T:=solve(-x^4+5*x^3+x^2-1,x);
```

$$T := RootOf(\_Z^4 - 5\,\_Z^3 - \_Z^2 + 1, index = 1), RootOf(\_Z^4 - 5\,\_Z^3 - \_Z^2 + 1, index = 2), RootOf(\_Z^4 - 5\,\_Z^3 - \_Z^2 + 1, index = 3), RootOf(\_Z^4 - 5\,\_Z^3 - \_Z^2 + 1, index = 4) \tag{16}$$

```
> allvalues([T]):
  allvalues([T])[1];
```

$$\frac{5}{4} + \frac{1}{12}\sqrt{3}\sqrt{\frac{83\left(2980 + 12\sqrt{60693}\right)^{1/3} + 2\left(2980 + 12\sqrt{60693}\right)^{2/3} + 104}{\left(2980 + 12\sqrt{60693}\right)^{1/3}}} \tag{17}$$

$$- \frac{1}{12}$$

$$\Bigg(\bigg(498\left(2980 + 12\sqrt{60693}\right)^{1}$$

$$/3\sqrt{\frac{83\left(2980 + 12\sqrt{60693}\right)^{1/3} + 2\left(2980 + 12\sqrt{60693}\right)^{2/3} + 104}{\left(2980 + 12\sqrt{60693}\right)^{1/3}}}$$

$$- 6\sqrt{\frac{83\left(2980 + 12\sqrt{60693}\right)^{1/3} + 2\left(2980 + 12\sqrt{60693}\right)^{2/3} + 104}{\left(2980 + 12\sqrt{60693}\right)^{1/3}}}\,\Big(2980$$

$$+ 12\sqrt{60693}\Big)^{2/3}$$

$$- 312\sqrt{\frac{83\left(2980 + 12\sqrt{60693}\right)^{1/3} + 2\left(2980 + 12\sqrt{60693}\right)^{2/3} + 104}{\left(2980 + 12\sqrt{60693}\right)^{1/3}}}$$

$$+ 2610\sqrt{3}\left(2980 + 12\sqrt{60693}\right)^{1/3}\bigg) \bigg/ \bigg(\left(2980 + 12\sqrt{60693}\right)^{1}$$

$$1/2$$

$$/3\sqrt{\frac{83\left(2980 + 12\sqrt{60693}\right)^{1/3} + 2\left(2980 + 12\sqrt{60693}\right)^{2/3} + 104}{\left(2980 + 12\sqrt{60693}\right)^{1/3}}}\bigg)\Bigg)$$

Maple peut aussi résoudre quelques **équations non polynomiales** simples.

```
> solve(exp(x)=1+x,x);
```

$$0 \tag{18}$$

Pour les équations plus compliquées, Maple peut trouver une partie des solutions, aucune des solutions, ou même des solutions qui sont fausses !

1.2 Résoudre un système d'équations

Pour résoudre un système d'équations, on utilise la commande *solve* avec la syntaxe : *solve({equations}, {variables})*
Maple retourne les solutions sous la forme {variable1=expression1, variable2=expression2,...}.

Exemple : le système {x+y+z=4, x+y-z=1, x-y-z=-3}.

```
> solve({x+y+z=4,x+y-z=1,x-y-z=-3},{x,y,z});
```

$$\left\{x=\frac{1}{2}, y=2, z=\frac{3}{2}\right\} \quad (19)$$

Ici, Maple trouve une solution unique.

Dans certains cas, on a une famille de solutions, qui s'expriment en fonction de certaines des variables.
Exemple : le système { x-3y=2,3x-9y=6}.

```
> solve({x-3*y=2,3*x-9*y=6});
```

$$\{x=3\,y+2, y=y\} \quad (20)$$

Maple a choisi de garder l'inconnue y comme paramètre, et a pu exprimer toutes les solutions en fonction de ce paramètre.

1.3 Résolution approchée

Lorsqu'on ne peut résoudre de manière exacte une équation (sans paramètre), on peut essayer la fonction *fsolve* pour obtenir des solutions *approchées*.

```
> fsolve(tan(sin(x))=1,x);
```

$$0.9033391108 \quad (21)$$

On peut spécifier un intervalle sur lequel on cherche les solutions :

```
> fsolve(x^5-7*x^4+3*x^2+15,x,0..10);
```

$$1.386962979, 6.931051639 \quad (22)$$

Par défaut, Maple cherche des solutions approchées réelles. On peut aussi lui demander des solutions approchées complexes :

```
> fsolve(x^5-7*x^4+3*x^2+15,x,complex);
```

$$-1.242272278, -0.03787117016-1.120100886\,I, -0.03787117016+1.120100886\,I, 1.386962979, 6.931051639 \quad (23)$$

2. Dérivation et intégration

Encore une fois, il s'agit de bien différencier la fonction f de l'expression f(x).

2.1 Dériver une expression

On dérive une *expression* à l'aide de la commande *diff*, par la syntaxe *diff(expression_en_variable, variable).*

```
> diff(1/(1+x^2),x);
```

$$-\frac{2\,x}{\left(1+x^2\right)^2} \quad (24)$$

Maple retourne alors une *expression* en la variable.

2.2 Dériver une fonction

On dérive une *fonction* (préalablement définie à l'aide de -> ou de unapply) par l'opérateur *D*.

```
> restart; f:=x->cos(x)+5*x^2-3;
```

$$f := x \to \cos(x) + 5\,x^2 - 3 \quad (25)$$

```
> g:=D(f);
```

$$g := x \to -\sin(x) + 10\,x \quad (26)$$

Maple retourne alors une *fonction*.

```
> g(Pi);
```

$$10\,\pi \quad (27)$$

2.3 Calculer une primitive

On utilise la commande *int*, dont la syntaxe est *int(expression_en_variable,variable)*. Le résultat est sans constante d'intégration. C'est une expression en la variable.

```
> int(tan(x),x);
```

$$-\ln(\cos(x)) \quad (28)$$

Parfois, Maple ne sait pas calculer exactement une primitive.

```
> int(exp(x)*cos(x)^n,x);
```

$$\int e^x \cos(x)^n \,dx \quad (29)$$

Vérifier qu'ici, si l'on donne une valeur précise à n, Maple sait mener le calcul.
Parfois, Maple exprime les primitives en fonction de fonctions dites spéciales :

```
> int(sin(x)/x,x);
```

$$\mathrm{Si}(x) \quad (30)$$

C'est la fonction sinus intégral (pour en savoir plus, consulter l'aide sur Si). Maple la connait, il sait par exemple en tracer le graphe :

```
> plot(Si);
```



2.4 Calculer une intégrale

Pour calculer l'intégrale de a à b de f, on utilise la commande *int* avec la syntaxe *int (expression_en_variable, variable=a..b)*.

```
> restart; int(exp(x)*cos(x)^3,x=0..Pi/2);
```

$$-\frac{2}{5}+\frac{3}{10}\,e^{\frac{1}{2}\pi} \tag{31}$$

Les bornes ne sont pas nécessairement des constantes. Elles peuvent être des indéterminées :

```
> int(x^2*cos(x),x=0..N*Pi);
```

$$N^2\pi^2\sin(N\pi)-2\sin(N\pi)+2N\pi\cos(N\pi) \tag{32}$$

Elles peuvent etre aussi +l'infini (noté *infinity*) ou -l'infini (noté -*infinity*).

```
> int(exp(-x^2)*cos(x),x=0..infinity);
```

$$\frac{1}{2}\sqrt{\pi}\,e^{-\frac{1}{4}} \tag{33}$$

3. Limites

Pour calculer des limites d'expressions en une variable, on utilise la commande *limit* avec la syntaxe *limit(expression_en_x,x=a)*. La valeur de x pour laquelle on cherche la limite de l'expression est a.

```
> restart; f:=x->(1-cos(x))/x^2 ;
```

$$f:=x\to\frac{1-\cos(x)}{x^2} \tag{34}$$

La fonction f n'est pas définie en 0, mais a une limite (finie) en 0 que Maple peut calculer :

```
> f(0);
  limit(f(x),x=0);
Error, (in f) numeric exception: division by zero
```

$$\frac{1}{2} \tag{35}$$

a peut valoir + l'infini et -l'infini (infinity et -infinity) :

```
> limit(ln(x),x=infinity);
```

$$\infty \tag{36}$$

```
> limit(1/x,x=0);
```

$$\textit{undefined} \tag{37}$$

Ici, Maple répond 'undefined' car la fonction x->1/x n'a pas de limite en 0. Par contre, elle a une limite à gauche et une limite à droite :

```
> limit(1/x,x=0,left); limit(1/x,x=0,right);
```

$$-\infty$$

$$\infty \tag{38}$$

Pour définir une suite dont on connaît le terme général, on procède comme pour une fonction "ordinaire" : par exemple, pour la suite u_n définie par u_n=cos(1/n) :

```
> u:=n->cos(1/n);
```

$$u := n \to \cos\left(\frac{1}{n}\right) \tag{39}$$

On peut alors utiliser la commande *limit* pour calculer la limite :

```
> limit(u(n),n=infinity);
```

$$1 \tag{40}$$

3.2. Continuité d'une fonction

Maple dispose de commandes pour étudier la continuité et la discontinuité des fonctions de **R** dans **R**. La commande *iscont* permet de tester la continuité d'une expression en *x* sur un intervalle donné. Le résultat donné par Maple est un booléen : *true* (vrai) si la fonction est continue sur l'intervalle, *false* (faux) si elle ne l'est pas. La syntaxe est *iscont(expression_en_x,x=a..b)* .

```
> restart; f:=x->1/(x-1);
  iscont(f(x),x=0..2);
```

$$f := x \to \frac{1}{x-1}$$

$$\textit{false} \tag{41}$$

Ici, f n'est pas continue sur l'intervalle 0..2 car elle n'est pas continue en 1. Attention, par défaut, Maple étudie la continuité sur l'intervalle *ouvert*]a,b[. Par exemple, :

```
> iscont(f(x),x=0..1);
```

$$\textit{true} \tag{42}$$

car *f* n'est pas continue au point 1, mais l'est sur l'intervalle ouvert en 1]0,1[. Il est possible de lui demander de travailler sur l'intervalle *fermé* [a,b] :

```
> iscont(f(x),x=0..1,'closed');
```

$$\textit{false} \tag{43}$$

La commande *discont* donne l'ensemble des points de discontinuité. Sa syntaxe est *discont* *(expression_en_x,x)*.

```
> discont(f(x),x);
```

$$\{1\} \tag{44}$$

```
> discont(1/sin(x),x);
```

$$\{\pi\ \_Z1\sim\} \tag{45}$$

(Quel est l'ensemble des points de discontinuité ? Essayez de comprendre la dernière réponse de Maple)